# Nock for Everyday Coders

# Tim Galebach ~timluc-miptev Hyperware

Abstract

Nock is not super complex, and most programmers can learn the basics of it rapidly. The mental model gained by learning Nock turns out to be very useful in learning Hoon and understanding Urbit. While the Urbit docs generally suggest to not worry about Nock, Nock is very simple and small. Most programmers will feel more comfortable in Hoon having learned Nock's basics. The goal of this tutorial is to explain Nock clearly in terms most programmers will relate to, to impart a feeling of confidence with very basic Nock, and to give a knowledge of Nock's idioms and big wins so that they carry over to learning Hoon.

This tutorial was originally published online by ~timluc-miptev in early 2020 and remains an excellent exposition of Nock's affordances.

# Contents

1

3

11

12

13

16

19

21	1	Gett	ing Started	2
22		1.1	Phases of Nock	3
23			What Is a Nock Interpreter?	3
24		1.3	How to Run Nock Code	4
25		1.4	Subject, Formula?	4
26			Evaluating Our First Nock Code	4

*Urbit Systems Technical Journal II:1* (2025): 1–46.

Nock's Simplest Functions, 0 and 1  Binary Tree Addressing	5 5 7 9 10 13 15 19 22 23 24 28 28 30 32
<ul> <li>0, the "Memory Slot" Function</li> <li>1, the "Quoter" Function</li> <li>4, the Incrementing Function</li> <li>The Cell-Maker (AKA the Distribution Rule)</li> <li>3, the Cell Detector, and 5, the Equality Tester</li> <li>2, the "Subject-Altering" Function</li> <li>Summary of Fundamental Opcodes</li> <li>An Aside About the 1 (Quoter Function)</li> <li>6, "If/Else" Conditional Branch</li> <li>7, the "Composition" Opcode</li> <li>8, the "Variable Push" Opcode</li> <li>9, Run a Stored Procedure Arm in a Core</li> <li>10, Replace a Memory Slot</li> </ul>	7 9 10 13 15 19 22 23 24 28 28 30
1, the "Quoter" Function 4, the Incrementing Function The Cell-Maker (AKA the Distribution Rule) 3, the Cell Detector, and 5, the Equality Tester 2, the "Subject-Altering" Function Summary of Fundamental Opcodes  An Aside About the 1 (Quoter Function) 6, "If/Else" Conditional Branch 7, the "Composition" Opcode 8, the "Variable Push" Opcode 9, Run a Stored Procedure Arm in a Core 10, Replace a Memory Slot	9 10 13 15 19 22 <b>23</b> 23 24 28 28 30
4, the Incrementing Function	100 133 155 199 222 233 244 282 283 30
The Cell-Maker (AKA the Distribution Rule) 3, the Cell Detector, and 5, the Equality Tester 2, the "Subject-Altering" Function Summary of Fundamental Opcodes	13 15 19 22 <b>23</b> 24 28 28 30
3, the Cell Detector, and 5, the Equality Tester 2, the "Subject-Altering" Function Summary of Fundamental Opcodes	15 19 22 23 24 28 28 30
2, the "Subject-Altering" Function Summary of Fundamental Opcodes	19 22 23 23 24 28 28 30
Summary of Fundamental Opcodes	22 23 24 28 28 30
mposite Opcodes  An Aside About the 1 (Quoter Function)	23 23 24 28 28 30
An Aside About the 1 (Quoter Function) 6, "If/Else" Conditional Branch	23 24 28 28 30
6, "If/Else" Conditional Branch	24 28 28 30
7, the "Composition" Opcode	28 28 30
8, the "Variable Push" Opcode	28 30
9, Run a Stored Procedure Arm in a Core 10, Replace a Memory Slot	30
10, Replace a Memory Slot	
-	32
Real Nock Code	
	33
Summary	35
erlude	35
Order of Operations	35
$\underline{11}$ , Hints and Side Effects for the Interpreter $$ .	36
Nock in Hoon	38
e Core as Design Pattern	38
Building a Core Manually	38
nclusion	45
etting Started	
	Order of Operations

The problem is, the programmer may already know that Nock code looks more like the below – just lists of numbers, with no symbols:

```
62 [6
63 [5 [0 6] [1 0]]
64 [0 7]
65 [9 4 [[0 2] [2 [0 6] [0 5]] [4 0 7]]]
66 ]
```

What gives? Which one is the "real" Nock?

#### 1.1 Phases of Nock

67

70

77

78

80

81

83

84

25

We are looking at two different things in the examples above:

- 1. Pseudocode for *how to interpret* Nock.
- 2. Code to be interpreted (written as lists of numbers).

The symbols and spec are pseudocode, not real Nock code.
They could just as easily be written in English, and they will
never be written down as actual Nock code and given to an
interpreter. They represent what an interpreter should do to
turn Nock code into interpreter instructions.

The lists of numbers are the actual Nock code. This is what you feed to an interpreter to get some result.

# 1.2 What Is a Nock Interpreter?

An interpreter can be a computer program, or it can be a human manually expanding Nock code into results. In both cases, the program and human have to know the Nock pseudocode in order to do the right thing with incoming Nock code.

So a Nock interpreter is any entity that takes Nock code as input, and gives a noun as output. A noun can be:

```
s6 :: a number
s7 782
s8 :: a cell (pair with two elements)
s9 [782 9872]
99 :: each element can itself be a pair
```

```
91 [782 [9872 89728]]
92 :: the above can be written as
93 [782 9872 89728]
```

#### 1.3 How to Run Nock Code

We will be expanding Nock pseudocode manually in the exam ples that follow, in effect acting as our own interpreter.

If we want to check that we're getting the right results from our manual interpretation, we need to run a Nock interpreter, such as the Urbit Dojo.

- 1. Start up a Dojo session on a fake ship.<sup>1</sup>
- 2. At the prompt, we can execute Nock using .\* dottar, e.g., .\* (NOCK\_SUBJECT, NOCK\_FORMULA).

# 1.4 Subject, Formula?

104 Let's keep this simple:

98

qq

100

103

105

106

109

- subject = an argument to a function
  - formula = the function

That's it. We'll see below how this works, going really slowly with examples.

# 1.5 Evaluating Our First Nock Code

OK, so the interpreter takes two arguments, a "subject" and a "formula". Both are nouns (a number or a cell). Let's run some insanely simple Nock code in the Dojo:<sup>2</sup>

```
113 > .*(42 [0 1])
114 42
```

<sup>&</sup>lt;sup>1</sup>Consult https://docs.urbit.org for details.

<sup>&</sup>lt;sup>2</sup>Code samples beginning with > are Dojo inputs (to wit, Hoon expressions).

In the above, 42 is our subject. [0 1] is our formula.

Formulas are always cells, and the first element of the cell is a number that you can think of as *the name of the function*.

In this case, our function name is 0, which is the memory slot function. It is always followed by 1 number, in this case 1, which is the number of the memory slot to fetch in the subject.

Whenever we look at Nock code, we want to ask:

- What is the subject (function argument)? In this case, it's 42.
- What is the formula (function)? In this case, it's [0 1].
- What value does that formula (function) produce when called on this subject (argument)? In this case, the return value is 42.

Why is the return value 42? How does this formula work?

# 2 Fundamental Opcodes

115

116

117

118

119

124

125

127

128

130

135

136

137

138

139

140

144

145

# 2.1 Nock's Simplest Functions, 0 and 1

The two most basic Nock functions are 0 address and 1 constant. The goal here is to get strong intuitions of what they do, how they handle edge cases, and how this relates to the Nock spec/pseudocode.

# 2.2 Binary Tree Addressing


Before getting started on Nock proper, we should understand how Nock and Hoon handle addresses in binary trees. If you already understand why memory slot 5 of [['apple' %pie] [0b1101 0xdad]] is %pie, you are good to go and can skip ahead to the Section 2.3.

Every noun in Nock can be thought of as a tree, which means we can give an exact number to access any position in the tree. This means that, no matter how big our subject (argument) is, we can yank a value out of any part of it.

The Nock specification defines noun tree addressing:

This permits the address to be defined within a particular subtree as well as within the overall tree (Figure 1).

Figure 1: A binary tree with labeled node addresses to several layers.



That is, how do you say which slot number you want from a given tree? We say that:

- The tree root is address 1.
  - The head of every node n is 2n.
  - The tail of every node n is 2n + 1.

Let's take an example tree to illustrate. In Nock cell form, the tree is:

```
162 [[4 5] [6 14 15]]
```

Diagrammatically, the tree looks like Figure 2.

- 1 is the address of the whole tree, [[4 5] [6 14 15]].
- 2 is the address of the left branch, [4 5].


158


159

163

164

165





- 3 is the address of the right branch, [6 14 15].
- 15 is the value 15.

167

168

169

170

184

185

Let's play around now in the Dojo Nock interpreter so that we can confirm this. In each example, our subject (argument) will be the tree [[4 5] [6 14 15]].

```
171
    :: formula (function) [0 1]: get the whole tree
172
    > .*([[4 5] [6 14 15]] [0 1])
173
    [[4 5] 6 14 15]
174
175
    :: formula (function) [0 2]: get the left branch
176
    > .*([[4 5] [6 14 15]] [0 2])
177
    [4 5]
178
179
    :: formula (function) [0 7]: get the subtree in slot 7
180
    > .*([[4 5] [6 14 15]] [0 7])
184
    [14 15]
183
```

# 2.3 0, the "Memory Slot" Function

The pseudocode for the  $\underline{0}$  opcode<sup>3</sup> is as follows:

```
186

<u>*</u>[a <u>0</u> b] /[b a]
```

<sup>&</sup>lt;sup>3</sup>From this point on, we will write Nock opcodes in special type so that they can be distinguished from other numbers in Nock. This reflects Hoon practice, which marks Nock rules as constants like %0. While not strictly necessary, it can be helpful when learning or coding Nock to see an explicit distinction.

190

191

192

193

194

195

196

197

198

/[b a] is pseudocode. In English, it means "a is a binary tree. Get the memory slot numbered b.

So if a were the tree [9 10], and b were 1, we'd get the memory slot 1 in the tree [9 10], which is just the tree itself.

Written in pseudocode, that's /[1 [9 10]]. We can also do /[2 [9 10]], which grabs memory slot 2, aka 9.

Let's look at some examples. I've put them first in Dojo form so that you can see how they run in that interpreter, and then I show the "human interpreter" in pseudocode below that.

# 2.3.1 Example: Retrieve the Subject

```
199
    :: get memory slot 1
200
    > .*([50 51] [<u>0</u> 1])
201
    [50 51]
202
    :: PSEUDOCODE -- replaces the Dojo's () with []
203
    *[[50 51] [0 1]]
204
    :: *[a @0@ b] -> a = [50 51], b = 1
205
    /[1 [50 51]]
206
    [50 51]
388
```

#### 209 2.3.2 Example: Retrieve the Head of the Subject

# 2.3.3 Example: A Crash

```
220

> .*([50 51] [0 [0 1]])

222 dojo: hoon expression failed

223 :: PSEUDOCODE

224 *[[50 51] [0 [0 1]]]

225 /[[0 1] [50 51]]
```

219

```
226 :: can't evaluate this--a memory slot must be a
227 number like 2, not a cell like [0 1]
228 CRASH
```

# 2.3.4 Summary of <u>0</u>

 $\underline{0}$  is everywhere in Nock, because "get something from a memory slot" really means "store stuff in a place and get it whenever I want," which is another name for creating variables. These memory slot numbers take the place of variable names.

If you're familiar with assembly or C, they're conceptually similar to memory pointers or registers in how Nock uses them. Keep in mind though, Nock is functional/immutable, so it doesn't update memory locations: it creates copies of data structures with altered values.

We've also seen that the memory slot function can't take just anything as the memory slot to fetch: it must receive an atom (number).

# 2.4 <u>1</u>, the "Quoter" Function

This is another really simple function. You can think of it as a "quoter": it just returns any value passed to it exactly as it is. It ignores the subject, and just quotes the value after it. Let's look at a couple examples.

```
248 > .*([20 30] [<u>1</u> 67])
249 67
250 :: [@1@ 2 587] is same as [@1@ [2 [587]]]
251 > .*([20 30] [<u>1</u> [2 [587])
252 [2 587]
```

It doesn't matter how much information is after the  $\underline{1}$ :  $\underline{1}$  is a dumb function that just returns it all.

The pseudocode for  $\underline{1}$  is:

```
*[a <u>1</u> b] b
```

In English, this means: "ignore the subject 'a', and just return everything after the  $\underline{1}$  exactly as it is.

Let's look at our first example, .  $\star$  ([20 30] [ $\underline{1}$  67]). The subject a is [20 30], so we ignore that. What's after the  $\underline{1}$ ? 67, so we return that.

In the second example, "everything after the <u>1</u>" is longer, but the same rule applies: the Nock interpreter just returns it exactly as it is, after stripping out unnecessary brackets.

# 2.4.1 Summary of 1

<u>1</u> is a simple function that just returns whatever is after it ("quoter" or "constant"). It's useful for quoting values that you want to use later in your Nock code.

# 2.5 4, the Incrementing Function

<u>0</u> and <u>1</u> are simple functions that don't have any nested behavior. Now we're going to move to a function that *does* have nested children, opcode <u>4</u>. In these examples, pay attention to how <u>4</u> operates on its arguments; we'll look at pseudocode and break down the examples in a moment.

```
> .*(50 [4 <u>0</u> 1])
277
     51
278
279
     > .*(50 [4 4 0 1])
280
     52
281
282
     > .*([100 150] [4 4 0 3])
283
     152
284
285
     > .*(50 [4 1 98])
286
     99
287
288
     > .*(50 [4 1 [0 2]])
289
     dojo: hoon expression failed
290
```

Here's  $\underline{4}$ 's pseudocode, juxtaposed with that of  $\underline{0}$  and  $\underline{1}$ :

```
292
293 *[a 4 b] +*[a b]
294 *[a 0 b] /[b a]
295 *[a 1 b] b
```

In English, this says "when we have subject a, function 4, 297 and Nock code b, first evaluate [a b] as [subject formula], 298 and then add 1 to the result."

If we contrast with  $\underline{0}$  and  $\underline{1}$ , we see that the right side has a \* symbol. This symbol means "evaluate the expression again in the Nock interpreter." 0 and 1 did not have this symbol, and that's why they couldn't evaluate nested Nock expressions.

#### Example: Walking Through Increment 2.5.1 304

299

300

301

302

303

327

Let's start by translating the Dojo's . \* (subject formula) to pseudocode of the form \*[a b], and then expand it line by 306 line: 307

```
> .*(50 [4 <u>0</u> 1])
308
309
    :: change to pseudocode
    *[50 [4 0 1]]
311
    :: move the 4 to the outside as +
    +*[50 [0 1]]
    :: expand the 0 opcode to the memory slot operator "/"
314
    +/[1 50]
315
    :: grabs the memory slot
316
    +(50)
317
    :: evaluate
    51
319
    The [a \ b] part of +*[a \ b] expands to:
320
321
    [50 [0 1]]
```

OK, this we know how to handle! It's just our 0 function, and 322 it wants the value in memory slot 1 of the subject. That's 50. 323

Now we know that \*[a b]=50, and we just have to add 1 324 to it (the + in +\*[a b]). That is 51, which is exactly what the 325 interpreter gave us. 326

#### Example: Walking Through Increment 2.5.2

This one is similar, we just have an extra 4. We again start by 328 translating the Dojo's . \* (subject formula) to pseudocode, 329 and then expand 330

```
> .*(50 [4 4 0 1])
331
     52
332
     *[50 [<u>4 4 0</u> 1]]
333
     :: the first 4 moves outside as a '+'
334
     +*[50 [<u>4</u> <u>0</u> 1]]
335
     :: 2nd 4 becomes a '+'...we have opcode 0 again!
336
     ++*[50 [0 1]]
     ++/[<u>1</u> 50]
338
339
     ++(50)
340
     +(51)
     52
341
```

### 2.5.3 Example: Walking Through Serial Increment

Here we again see lots of <u>4</u>s applied consecutively, and we also see how we can yank values out of a more complicated subject and manipulate them. Notice how the subject is a cell, not an atom. The rest is the same as in the previous example.

```
> .*([100 150] [4 4 0 3])
347
    152
348
    a \text{ (the subject)} = [100 150]
349
    *[[100 150] [4 4 0 3]]
350
    +*[[100 150] [4 0 3]]
    :: Now we've extracted all the increments,
352
    :: so we just grab the value at memory slot 3
353
    ++*[[100 150] [0 3]]
354
    ++/[3 [100 150]]
355
    ++(150)
356
    +(151)
357
    152
358
```

# 2.5.4 Example: Walking Through Incrementing a Constant

In the below example, we see how we can use the quote/constant function <u>1</u> to generate the value 98 and increment it. We ignore the subject 50 completely.

```
363 → .*(50 [<u>4</u> <u>1</u> 98])
364 99
365 *[50 [<u>4</u> <u>1</u> 98]]
366 :: formula is the "quoter" function
```

```
367 +*[50 [<u>1</u> 98]]
368 +(98)
369 99
```

## 2.5.5 Example: A Crash When Incrementing a Cell

Just as opcode <u>0</u> had values it couldn't handle (non-atoms), so opcode <u>4</u> needs the nested value inside it to evaluate to an atom.

```
> .*(50 [4 1 [0 2]])
    dojo: hoon expression failed
374
    *[50 [4 1 [0 2]]]
    :: OK cool, the nested value is a 1 opcode
376
    +*[50 [1 [0 2]]]
373
    :: 1 ignores the subject (50) and just returns [0 2]
379
    +([0 2])
    :: [0 2] isn't an atom, so how can we increment it????
380
    :: We can't, so we crash.
381
    dojo: hoon expression failed
388
```

# 383 2.5.6 Summary of <u>4</u>

In these examples, we've seen that function  $\underline{4}$  can be called as many times in a row as we want. At the end of those calls, it always ends up incrementing a number that either is yanked from the subject (memory slot function  $\underline{0}$ ) or quoted as it is (quote function  $\underline{1}$ ).

# 2.6 The Cell-Maker (aka the Distribution Rule)

The Nock interpreter is allowed to return nouns, which are atoms (positive numbers) or cells (pairs of nouns). What have our functions/opcodes been returning so far?

- <u>0</u>: atoms or cells, depending what's in the memory slot that we yoink.
- <u>1</u>: atoms or cells, depending on what we quote.
- <u>4</u>: just atoms.

393

394

398

399

400

401

402

403

404

405

406

407

408

409

410

411

412

413

416

419

420

123

424

But what if my subject was [51 67 89], and I wanted to increment every value and return that as [52 68 90]? How can I do that when it's a cell, and 4 only seems able to return atoms?

The answer is something that the Nock docs call the "distribution rule" or "implicit cons" (hello, fellow LISPers!), but that I find easiest to think of as the "Cell-Maker Rule".

#### 2.6.1 A Quick Detour into Nock Formulas

We haven't talked much yet about what values are legal to feed into the Nock interpreter (the .\*(subject formula) function in the Dojo). So far, we've only been using formulas that start with numbers (our functions/opcodes 0/1/4).

Let's now fully solidify our understanding of what's a legal formula by taking a quick look at the three possible cases. (The format is. \*(subject formula).)

 The formula is a cell starting with an opcode. We know this is OK.

```
414 > .★(50 [<u>0</u> 1])
415 50
```

• The formula is just an atom (o). This is not valid.

```
417 > .★(50 0)
418 dojo: hoon expression failed
```

• Finally, we try a formula that is a cell starting with an atom. This looks invalid, but—it works!

```
421 > .*(50 [[<u>0</u> 1] [<u>1</u> 203]])
422 [50 203]
```

So apparently a formula cell can start with a cell. The Cell-Maker rule is:

```
425

*[subject [formula-x formula-y]]→

427

[*[subject formula-x] *[subject formula-y]]
```

In our example above, formula-x is [0 1], and formula-y is [1 203]. They each evaluate individually against the subject, and the end result is a cell.

We can make as many cells in a row as we want:

```
433 → .★(50 [[0 1] [1 203] [0 1] [1 19] [1 76]])
434 [50 203 50 19 76]
```

We can put any operation inside each cell:

```
<sup>436</sup> → .★([19 20] [[<u>0</u> 1] [<u>1</u> 76] [<u>4</u> <u>4</u> <u>0</u> 3]])

<sup>437</sup> [[19 20] 76 22]
```

432

435

441

448

449

450

451

452

If we take the returned collection [[19 20] 76 22] in order, we can write in English how they connect to our collection of formulas that we passed:

- [<u>0</u> 1]: get memory slot 1
- [<u>1</u> 76]: return the quoted value 76
- [ $\frac{4}{2}$   $\frac{6}{2}$  3]: increment twice the value in memory slot 3 (20)

So we can pass one small subject ([19 20]) and make an arbitrarily long collection of values from it, using any functions we want. Cell-Maker FTW!

# 2.7 3, the Cell Detector, and 5, the Equality Tester

Now we come to functions/opcodes  $\underline{3}$  and  $\underline{5}$ , which are pretty straightforward after we've seen how  $\underline{4}$  and the Cell-Maker work. Functions  $\underline{3}$  and  $\underline{5}$ , like  $\underline{4}$ , allow nested evaluation. Let's put all their pseudocode definitions together to compare:

```
453

:: function/opcode 3

455 *[a 3 b] ?*[a b]

456 :: function/opcode 4

457 *[a 4 b] +*[a b]

458 :: function/opcode 5

460 *[a 5 b c] =[*[a b] *[a c]]
```

462

463

464

465

466

468

469

470

471

472

473

485

First of all, notice how the right side of all these "equations" has the evaluation operator, \*. This means that these functions can have nested formulas, since they keep evaluating all the way down.

There are some new pseudocode symbols here that we need to translate into English. We already know +: "increment the value after this". Now we also see:

- ?: "check whether the value after this is a cell. Return o
  if yes, 1 if no".4
- =: "first run the function in b with subject a as the argument, and same for the function in c. If the results are equal, return 0, if not, return 1.

#### 2.7.1 Example: Not a Cell

```
474
     > .*(50 [<u>3</u> <u>0</u> 1])
475
476
     :: PSEUDOCODE OF THE ABOVE
477
     *[50 [<u>3</u> <u>0</u> 1]]
478
     ?*[50 [<u>0</u> 1]]
479
     ::get memory slot 1 of the subject
480
     ?(50)
481
     :: is 50 a cell? No, so return 1
482
483
```

#### 2.7.2 Example: A Cell

```
486
    > .*([50 51] [3 0 1])
487
488
    :: PSEUDOCODE OF THE ABOVE
489
490
    *[[50 51] [3 0 1]]
491
    ?*[[50 51] [0 1]]
    ::get memory slot 1 of the subject: [50 51]
492
    ?([50 51])
    :: is [50 51] a cell? Yes, so return 0
494
485
```

 $<sup>^4</sup>$ Nock conventionally explains that there is one "true" case and (potentially) many "false" cases, thus o/true and 1/false.

## 2.7.3 Example: Nested Cell Evaluation with 4

497

515

```
498
    > .*([50 51] [4 4 3 0 1])
499
500
    :: PSEUDOCODE OF THE ABOVE
501
    *[[50 51] [4 4 3 0 1]]
    +*[[50 51] [4 3 0 1]]
503
    :: whatever comes out of the 3 function,
    :: we're gonna increment it twice
505
    ++*[[50 51] [3 0 1]]
506
    :: down to just fetching memory slot 1
507
    ++?*[[50 51] [0 1]]
508
    ++?([50 51])
509
    :: is [50 51] a cell? Yes, so return 0
    ++(0)
511
512
    +(1)
5‡<u>3</u>
```

# 2.7.4 Example: Check Multiple Cases of Cells

```
516
    > .*([[50 51] 52] [[<u>3 0</u> 2] [<u>3 0</u> 3]])
517
    [0 1]
    *[[[50 51] 52] [[<u>3 0</u> 2] [<u>3 0</u> 3]]]
519
    ?[*[[50 51] 52] [0 2]]
520
        *[[50 51] [0 3]]]
523
    :: yank memory slots 2 and 3
522
    ?*[[50 51] 52]
523
    :: first is a cell, second is not
    [0 1]
525
```

# 527 2.7.5 Example: Equality Test

Because 5 compares the results of 2 formulas, it *always* makes
 2 inner evaluations of the subject. It's similar to Cell-Maker in
 this way.

```
531

> .*([50 51] [5 [0 2] [0 2]])

533  0

534  :: PSEUDOCODE

535  *[[50 51] [5 [0 2] [0 2]]]
```

554

```
536 :: factor out the = 

537 = [*[[50 51] [0 2]] *[[50 51] [0 2]]] 

538 :: get memory slot 2 twice 

539 = (50 50) 

\frac{540}{40} 0
```

# 2.7.6 Example: Comparing Two Unequal Values

```
543
    > .*([50 51] [5 [0 2] [0 3]])
544
    1
545
    :: PSEUDOCODE
546
    *[[50 51] [5 [0 2] [0 3]]]
547
    :: factor out the =
548
    =[*[[50 51] [<u>0</u> 2]] *[[50 51] [<u>0</u> 3]]]
549
    :: get memory slot 2 and memory slot 3
550
    =(5051)
    1
553
```

## 2.7.7 Example: Compare Values with Function Calls

```
> .*([50 51] [5 [4 0 2] [0 3]])
556
557
     :: PSEUDOCODE
558
     *[[50 51] [<u>5</u> [<u>4</u> <u>0</u> 2] [<u>0</u> 3]]]
559
     :: factor out the =
569
     =[*[[50 51] [<u>4 0</u> 2]]
561
         *[[50 51] [<u>0</u> 3]]]
     :: factor out the +
563
     =[+*[[50 51] [0 2]]
564
          *[[50 51] [<u>0</u> 3]]]
565
     :: get memory slots 2 and 3
566
     =[+50 51]
567
     :: evaluate the +
    =([51 51])
569
     0
579
```

# 2.7.8 Example: Compare Cells with Function Calls

```
573
    > .*([99 99] [5 [1 [99 99]] [0 1]])
574
575
    :: PSEUDOCODE
576
    *[[99 99] [5 [1 [99 99]] [0 1]]]
577
    =[*[[99 99] [<u>1</u> [99 99]]]
578
      *[[99 99] [<u>0</u> 1]]]
579
    :: first cell is the result of quoter function
    :: second cell is the result of memory fetch
581
    =[[99 99] [99 99]]
582
583
    :: true
    0
584
```

## 2.7.9 Summary of 3 and 5

586

591

592

593

594

596

597

598

599

600

601

602

606

607

608

These along with <u>4</u> are known as the "axiomatic" functions.
They exist to implement definitional logic for the Nock specification. In particular, the cell check and the equality check provide for structural analysis of nouns.

# 2.8 2, the "Subject-Altering" Function

In all our examples so far, the subject has been defined at the start when we call the interpreter, and never changes. But what if we want to change the subject?

A different subject? Why would we want that? Here's an easy example. Say you found the following piece of Nock code on the interwebz:

```
[8 [1 0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0 2]
[4 0 6] 0 7] 9 2 0 1]
```

This is the code for a Nock function that expects a subject that is an atom, and decrements that subject by 1. You can actually enter it in the Dojo right now:

```
603 > .*(100 [8 [1 0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0
604 2] [4 0 6] 0 7] 9 2 0 1])
605 99
```

This works! You do not have to understand the code right now; just try entering different numbers instead of 100 to see the program working.

But now imagine that I have a Nock program with a different subject 610

```
.*([50 51] some-formula)
611
```

609

612

613

614

616

617

618

622

623

624

625

626

627

629

631

632

633

634

And somewhere inside some-formula, I want to decrement the number 51. I can't pass [50 51] as the subject to my decrementer code above though, because that's a cell, and it expects just an atom (a number).

#### Subject Altering to the Rescue 2.8.1

The function/opcode 2 is designed to handle this problem for us. First the pseudocode:

```
619
    *[a 2 b c]
                *[*[a b] *[a c]]
620
```

2 expects 2 formulas after the subject a: b and c. With those, it:

- 1. runs formula b against the subject to set up a new environment/subject derived from the subject
- 2. runs formula c against the subject to prepare a 2nd function
  - 3. run that 2nd function against the new environment/subject from step (1)

Note that the pseudocode for 2 has nested "s.". 630

```
*[*[a b] *[a c]]
```

The two inner \*s run steps (1) and (2), and the outer one, around the whole expression, runs step (3).

#### 2.8.2 Example: Change the Subject and Call a Constant Value

```
635
    > .*([50 51] [2 [0 3] [1 [4 0 1]]])
636
637
    52
    :: PSEUDOCODE
638
    *[[50 51] [2 [0 3] [1 [4 0 1]]]]
639
    :: separate b and c to each run against the subject
640
         (steps 1 and 2)
641
```

```
*[*[[50 51] [0 3]] *[[50 51] [1 [4 0 1]]]]
    :: after steps 1 and 2, we have a new subject, 51!
643
    :: note how we're back in normal *[subject formula]
        form
645
    *[51 [4 0 1]]
646
    :: apply the 4 function as we're used to
647
    +*[51 [0 1]]
    :: grab 51 from memory slot 1
649
650
    +(51)
    52
651
```

#### 2.8.3 Example: Grab a Block of Code from the Subject and Run It

653

654

673

674

675

676

677

Think of this as grabbing a "stored procedure" from the subject.

```
655
    > .*([[4 0 1] 51] [2 [0 3] [0 2]]])
656
    52
657
    :: PSEUDOCODE, subject is [[4 0 1] 51]
658
    *[[[4 0 1] 51] [2 [0 3] [0 2]]]
659
    *[*[[[4 0 1] 51] [0 3]
660
      *[[[4 0 1] 51] [0 2]]]]
    :: step 1 gets memory slot 3, step 2 grabs memory
662
         slot 2
663
    *[51 [4 <u>0</u> 1]]
664
    :: looks like a normal 4 opcode to me!
665
    +*[51 [<u>0</u> 1]]
666
    :: grab memory slot 1
667
    +(51)
668
    52
669
```

# 2.8.4 Example: Back to Our Motivating Case

Remember our decrementing block of code that we couldn't use when the subject was [50 51], instead of just an atom? Opcode 2 makes handling that issue a piece of cake.

We simply use  $\underline{2}$  to transform our subject into an atom, and use  $\underline{1}$  to quote the decrement block of code before it evaluates in step (3).

```
678 

679 

- .*([50 51] [2 [0 2] [1 [8 [1 0] 8 [1 6 [5 [0 7] 4 0]

680 

6] [0 6] 9 2 [0 2] [4 0 6] 0 7] 9 2 0 1]]])
```

```
681 49
682 :: PSEUDOCODE (subject is [50 51])
683 :: ("decrement-formula" substituted for clarity)
684 *[[50 51] [2 [0 2] [1 decrement-formula]]]
685 *[*[[50 51] [0 2]] *[[50 51] [1 decrement-formula]]]
686 *[50 decrement-formula]
687 ::decrement-formula has the atom subject that it wants
688 49
```

## 2.8.5 Summary of 2

For those who know a bit of Hoon, the second example above is rather similar to calling an arm that produces a gate, and then running the gate. Most of Hoon runs this type of stored-procedure plus subject-altering Nock, and it all uses opcode 2 at its base.

And for those who know Hoon, [[4 0 1] 51] should already be looking a lot like [battery payload]... that's not a coincidence. We're starting to see the first glimpses of how Hoon's cores, arms and subjects/subject mutations flow out of the fundamental structure of Nock.

We also see how Hoon/Nock lend themselves well to throwing around chunks of code, and adjusting the subject as necessary to create the correct subject/environment against which to run that code.

# 2.9 Summary of Fundamental Opcodes

You now have seen all the fundamental functions/opcodes in Nock. In the next part, I'll introduce the remaining functions, which no longer need pseudocode: we have enough scaffolding now to build the rest of Nock from Nock itself. Instead, these new opcodes will just be shortcuts/macros/code expansions building on opcodes 0-5.

We'll also start to connect Nock to Hoon, and see how most of the fundamental (and slightly weird) features of Hoon flow directly from Nock's structure, and make a lot more sense in combination with it.

# 3 Composite Opcodes

A word of encouragement: we're done with the hard part now. Every Nock function we learn in this section will be built from pieces in the preceding section.

None of these new functions are *necessary* to make Nock work. All of them except Nock <u>10</u> and <u>11</u> are syntactic sugar that is part of the Nock definition, and must be built into every correct Nock implementation. If you have seen macros or code expansions in other languages, that's another word for what's happening here.

Nock  $\underline{10}$  makes it easier to replace a memory value somewhere in a tree, and Nock  $\underline{11}$  allows passing hints to the interpreter.

One point of clarity: this syntactic sugar/code-expansion system is *not* extensible. This means that you can't invent your own Nock opcodes and still have that language be Nock; you're making a higher-level language on top of Nock at that point.

In fact, that's not a bad way to think of Hoon: it's a higher-level language that adds missing syntatic sugar/macros and human-readable names to Nock. (That's not the whole story, but it's a decent mental peg to initially hang Hoon on.)

Nock is intentionally very, very small, such that you can always walk through and analyze what is happening in a block of code if you know Nock's opcodes.

# 3.1 An Aside About the <u>1</u> (Quoter Function)

You're going to see the  $\underline{1}$  opcode (the "Quoter") appear in a lot of examples below for a simple reason: the  $\underline{6}$ - $\underline{9}$  opcodes expect formulas in a lot of places, not just atoms. And, as we've seen already, formulas have to be cells.

Whenever a formula is required, but you really just want to return a number, you use the quoter function.

# 3.1.1 Example: Incrementing a Constant

763

764

766

767 768

772

778

779

```
750 :: but 4 expects a formula after it, so we use [1 5]
751 > .*(0 [4 1 5])
753 6
```

# 3.1.2 Example: Comparing a Memory Slot to a Constant Value

```
755
756 :: we grab memory slot 2
757 :: then it has to be compared to the result of a
758 formula
759 :: so we just use the formula [1 23] to return 23
760 > .*([23 45] [5 [0 2] [1 23]])
762 0
```

# 3.2 6, "If/Else" Conditional Branch

The remaining Nock opcodes are "sugar". This means that the functions in this next part will use only  $\star$  and Nock code in their pseudocode. For example, here is the code for  $\underline{6}$ , the "If/Else" function.

```
758 *[a 6 b c d] *[a *[[c d] 0 *[[2 3] 0 *[a 4 4 b]]]]
```

The prose explanation of what's happening is straightforward:

- Evaluate b against the subject a (\*[a b]) to see whether
   it's 0/true or 1/false.
- 2. If b equals 0/true, run formula c against subject a.
- 3. If b equals 1/false, run formula d against subject a.
  - 4. If b is not equal to 0 or 1, the code crashes, for reasons we'll see in the code explanation.

#### 3.2.1 Code Explanation (Way More Fun)

OK, so that's the English version. The code explanation (the right side of the definition) is *really fun* now that we know the basic Nock opcodes.

The pseudocode has four nested '\*[subject formula]'s, so
I'm going to unwrap those to the bottom, and then build it up
again. The layers are, in order:

- 1. \*[a ...], i.e. subject a evaluated against that long formula starting with \*[[c d]  $\underline{0}$  ...].
- 2. \*[c d], i.e. subject c evaluated against the formula starting with [[2 3 ...].
- 790 3. \*[2 3], i.e. subject 2 evaluated against the lowest-level
   791 formula.
  - 4. Finally, subject a evaluated against formula [4 4 b].

#### 3.2.2 Step 4

792

793

797

807

ឧกឧ

മവമ

810

811

812

813

814

815

Remember, our English explanation was "see if \*[a b] is true or false, and do different actions depending on that. Step Four is that check.

Let's say we have a as 59, and b as the quoted value  $\underline{0}$ /true.

```
\star [a \underline{4} \underline{4} b]
798
     :: a: 59
799
     :: b: [1 0]
800
     *[59 <u>4 4</u> [<u>1</u> 0]]
801
     :: substitute out the two increment operators
     ++*[59 [1 0]]
803
     :: ignore subject, return quoted value 0
804
     ++(0)
805
     2
806
```

In summary, because  $\star$ [a b] evaluated to  $\underline{0}$ /true, we get the number 2. If  $\star$ [a b] had been  $\underline{1}$ /false, we'd get 3 (because we'd evaluate  $\star \star \star$  (1)).

What the heck? Why are we getting  $\underline{2}$  or  $\underline{3}$  back? How does that help us?? Well, Nock uses that returned  $\underline{2}$  or  $\underline{3}$  as a *memory slot*, and executes the code in that memory slot.

# 3.2.3 Step 3

Now we go up to step 3, with the subject [2 3] evaluated against our return value from step 4. Let's imagine that 2 had

#### been returned: 816

822

823

824

825

826

827

828

829

835

842

843

```
:: remember, "result-of-step-4" was 2
817
    *[[2 3] 0 result-of-step-4]
818
    *[[2 3] 0 2]
    :: get memory slot 2
820
823
```

This grabs memory slot 2 if \*[a b] was true, and memory slot 3 if \*[a b] was false.

Wait, isn't this redundant? We are just using our 2 or 3 generated in step 4 to generate a 2 or a 3. Seems dumb.

The answer is that we make sure that a crash happens if \*[a b] yields any answer other than 0/true or 1/false. If \*[a b] returned 10, for example, we'd have the following code in step 3:

```
*[[2 3] 0 10]
830
    :: there's no slot 10
831
    CRASHII
832
```

This is exactly what we want: the program crashes unless 833 we are doing a loobean<sup>5</sup> test that returns a 0 or 1 (converted to 834 an indicial 2 or 3) in step 4.

#### 3.2.4 Step 2 836

We now have our validated 2 or 3 to plug into step 2. Let's 837 imagine c is the simple formula [0 1] and d is [1 203]. 838

```
:: if step 3 returned "2" (true)
839
     *[[[<u>0</u> 1] [<u>1</u> 203]] <u>0</u> 2]
     [0 1]
841
```

Not much to see here: we just grab memory slot 2 or 3 depending on whether our initial b was true or false.

#### 3.2.5 Step 1 844

And now we're back at the top level, where we just use whichever 845 formula we yoinked in step 2 and run it against a.

<sup>&</sup>lt;sup>5</sup>A boolean truth value but with the typical truth values of o and 1 reversed to o/true and 1/false.

```
*[a formula-from-step-2]

** * [a formula-from-step-2]

** :: let's say we returned formula [0 1]

** our original a, from step 4, was 59

** [59 [0] 1]]

** 59
```

### 3.2.6 Example: Code Expansion of 6

852

```
\rightarrow .*(1 [6 [0 1] [0 1] [4 0 1])
854
     :: PSEUDOCODE
855
     :: *[a *[[c d] 0 *[[2 3] 0 *[a 4 4 b]]]]
856
     *[1 *[[[0 1] [4 0 1]] 0 *[[2 3] 0 *[1 4 4 [0 1]]]]]
     :: factor out the 4 opcodes
858
     *[1 *[[[0 1] [4 0 1]] 0 *[[2 3] 0 ++*[1 [0 1]]]]]
     :: b evaluates to 1 (yank memory slot 1)
860
     \star [\underline{1} \star [[\underline{0} \ 1] \ [\underline{4} \ \underline{0} \ 1]] \ \underline{0} \ \star [[2 \ 3] \ \underline{0} \ ++(1)]]]
861
     :: evaluate the two increments
862
     *[1 *[[[0 1] [4 0 1]] 0 *[[2 3] 0 3]]]
868
     :: get memory slot 3 from [2 3]
864
     *[1 *[[[0 1] [4 0 1]] 0 3]]
     :: [0 3] means get memory slot 3 from the subject
866
           (formula [4 0 1])
867
     *[1 [4 0 1]]
868
     :: factor out the 4 opcode
869
     +*[<u>1</u> [<u>0</u> 1]]
870
871
     +(1)
873
```

#### 3.2.7 Summary of 6

874

875

876

877

878

879

880

881

882

883

Now I'm going to make you a little sad. Most Nock interpreters don't do this whole awesome code expansion. They just see  $\underline{6}$ , and implement an if/else check with a crash if  $\star [a \ b]$  isn't a loobean.

However, the pattern of storing chunks of code in memory and pulling them out when you want them is the most important part of Nock. In fact, those of you who know Hoon are probably already seeing the makings of the "core" code pattern.

# 3.3 7, the "Composition" Opcode

 $\underline{7}$  is so simple we barely need to spend time on it. All it does is create a new subject/environment (using  $\underline{2}$ ), and immediately runs a formula against that new subject. Let's compare it to  $\underline{2}$ :

```
888
889 ::opcode 7
890 *[a 7 b c] *[*[a b] c]
891
892 :: opcode 2
888 *[a 2 b c] *[*[a b] *[a c]]
```

This is almost exactly the same except with c instead of  $\star$  [a c]. This means that you can write a "subject-changing" formula in b, and then run a simple function against that in c.

# 3.3.1 Example: Opcode 2 vs. Opcode 7

```
899
900
    :: using opcode 2
    > .*([23 45] [2 [0 3] [1 4 0 1]])
901
    46
902
903
    :: using opcode 7 -- we get to remove the quoter
904
         opcode "1"
905
    :: wow amazing /sarcasm
906
    > .*([23 45] [7 [0 3] [4 0 1]])
907
    46
888
```

Opcode  $\underline{7}$  is clearly trivial, so why does it exist? It allows clean expression of function composition: this is really c(b(a)), where a (the subject) is the initial argument, and b and c are functions. This pattern comes up a ton, and it's nice to not use  $\underline{1}$ s everywhere.

# 3.4 8, the "Variable Push" Opcode

If you're ever writing Nock and think "how can I add a new variable to the subject?", opcode  $\underline{8}$  is what you want. The new variable can be based on either the existing subject, or be a new value you add on.

This says to run \*[a b], and then make that the head of a new subject, with the old subject a as the new subject's tail.

# 3.4.1 Example: Add Variable as a Copied Value from an Existing Subject

In English, the below code first yanks the variable from memory slot 3, copies it to the head of a new subject, and then increments the value in that new head.

```
> .*([67 39] [<u>8</u> [<u>0</u> 3] [<u>4</u> <u>0</u> 2]])
     40
932
     :: PSEUDOCODE
933
     *[[*[[67 39] [0 3]] [67 39]] [4 0 2]]
934
     :: yanks mem slot 3 and pins it to the front of the
935
          old subject
936
     *[[39 [67 39]] [<u>4 0</u> 2]]
937
     +*[[39 [67 39]] [<u>0</u> 2]]
938
     +(39)
939
     40
949
```

## 3.4.2 Example: Add Variable as a New Value

```
943

> .*([67 39] [8 [1 0] [4 0 2]])

945 1

946 :: PSEUDOCODE

947 *[[*[[67 39] [1 0]] [67 39]] [4 0 2]]

948 *[[0 [67 39]] [4 0 2]]

949 +*[[0 [67 39]] [0 2]]

950 +(0)

951 1
```

Why do we want this? In the first example, we pin a copy of a value so that we can manipulate it without changing the original. In the second example, we add a  $\underline{0}$  to the front; maybe we want to increment it until some condition is met?

The above should be starting to feel *very* Hoon-ish: we're a minor code transform away from pinning new values to the head of a payload.

966

967

970

# 3.5 9, Run a Stored Procedure Arm in a Core

We're almost at full Hoon now, although still at a very low/raw
 level. 9 looks a little complicated...

```
963

*[a <u>9</u> b c] *[*[a c] <u>2</u> [<u>0</u> 1] <u>0</u> b]
```

... but in English, this is just saying:

- Use formula c to make a new subject from a (\*[a c]).
- 2. Grab the formula located at memory slot b in that new subject.
  - 3. Run that formula against the new subject \*[a c].

The fact that the above description has the words "make a new subject" tells us right away that it's syntactic sugar for opcode 2, and that's indeed what we see in the pseudocode.

# 974 3.5.1 Example: Using 9 to Run an Increment Arm

The initial expression here likely looks cryptic, but if you follow the pseudocode, it will become clear.

To stay oriented, remember that, if we're thinking of  $\underline{9}$  as  $\star$  [a 9 b c], then

```
979 a: 45
980 b: 2
981 c: [[<u>1 4 0</u> 3] <u>0</u> 1]
```

```
982
      > .*(45 [<u>9</u> 2 [<u>1</u> <u>4</u> <u>0</u> 3] <u>0</u> 1])
983
      46
984
      :: PSEUDOCODE
985
      *[45 [9 2 [<u>1 4</u> 0 3] [0 1]]]
986
      *[*[45 [<u>1</u> <u>4</u> <u>0</u> 3] [<u>0</u> 1]] 2 [<u>0</u> 1] <u>0</u> 2]
987
      :: new subject is [[4 0 3] 45]
988
      :: that is a formula to increment mem slot 3 in the
989
      :: head; 45 in the tail
ggn
      *[[[4 <u>0</u> 3] 45] <u>2</u> [<u>0</u> 1] <u>0</u> 2]
991
      :: now we expand opcode 2
992
      *[*[[<u>4</u> <u>0</u> 3] 45] <u>0</u> 1]
993
         *[[[4 <u>0</u> 3] 45] <u>0</u> 2]]
994
```

```
995 :: mem slot 2 of the subject becomes the new formula 996 \star [[4 0 3] 45] 4 0 3] 997 +\star [[4 0 3] 45] 0 3] 998 :: grab mem slot 3 999 + (45) 46
```

We start above with a subject that is not a core; it's just the atom 45. The code for c is then:

```
1004 [[1 4 0 3] [0 1]]
```

This formula uses the Cell Maker (Distribution Rule) to insert  $[4 \ 0 \ 3]$  as the head of the new subject, and puts mem slot 1 of the old subject as the tail, so we get a new subject of  $[4 \ 0 \ 3] \ 45$ .

What we do next is use b (here, 2) to select memory slot 2 from that new subject. Memory slot 2 is a formula:  $[\underline{4} \ \underline{0} \ 3]$ . We run that formula against the new subject.

# 3.5.2 Key Point/Possible Confusion

When I first saw  $\underline{9}$ , I thought: "why didn't they just use  $\underline{2}$  to make the transformed subject? Why is there an extra step to use  $[\underline{0} \ 1]$  to pull the new \*[a c] subject? Why can't we do \*[a  $\underline{2}$  c  $[\underline{0}$  b]]?

The answer is that we actually use the  $\star$  [a c] subject *twice*:

- We extract from it the formula located at memory slot b.
- 2. Then we run that formula against the \*[a c] subject.

If we just did  $\star$  [a 2 c [0 b]], then [0 b] would try to look up the b memory slot in a, NOT  $\star$  [a c]. So 9 gives us one extra step to set up the core itself.

#### 3.5.3 How to Think of 9

I like to think of 9 as having three parts:

1. c: our formula to set up the subject, making a new subject.

1028

1032

1035

1047

1048

1049

1050

- 2. b: the memory slot in our new subject where an arm is.
  - 3. Run the arm located at b against the new subject.

You can think of this as setting up a subject, pulling a stored procedure from it, and then running that procedure against the subject.

# 3.6 10, Replace a Memory Slot

Before explaining <u>10</u>, we need to introduce a new operator, #.

# is the "edit" operator. It has the form

```
#[mem-slot new-val target-tree]
```

It replaces the memory slot mem-slot in target-tree with new-val.

```
1040 :: Example
1041 #[2 [4 5] [99 88 77]]
1042 [[4 5] 88 77]
```

1043 The pseudocode for <u>10</u> is:

```
1044
†<u>|}∱</u> *[a <u>10</u> [b c] d] #[b *[a c] *[a d]]
```

In English, this first calculates \*[a c] and \*[a d], and then replaces memory slot b in the latter with the result of the former.

# 3.6.1 Example: Using <u>10</u> to Replace a Memory Slot

```
1051
      > .*(50 [<u>10</u> [<u>2</u> [<u>0</u> 1]] [<u>1</u> 8 9 10]])
1052
1053
      [50 9 10]
          PSEUDOCODE
1054
      *[50 [10 [2 [0 1]] [1 8 9 10]]]
     #[<u>2</u> *[50 <u>0</u> 1] *[50 <u>1</u> 8 9 10]]
1056
     #[2 /[1 50] [8 9 10]]
1057
     #[2 50 [8 9 10]]
1058
           expanded as far as we can, now do the edit
1059
           replace mem slot 2 of [8 9 10] with the value 50
1060
     [50 9 10]
1869
```

We will defer discussion of opcode <u>11</u> to a later section because it is not part of strict Nock semantics.

1063

1064

#### 3.6.2 Towards Hoon

1065

1072

If "grab a chunk of code from a subject and then run it against the subject" sounds a lot like getting an arm from a core in
Hoon, that's because it is. Nock opcode 9 is hand-in-glove with Hoon's core/arm structure. The "new subjects" created by

\*[a c] are cores, and the things selected from memory slots by b are arms.

#### 3.7 Real Nock Code

To finish this off and take your new powers for a spin, let's look at some real Nock code from the wild. I got the below example from the Dojo. It's a mold: a function that takes a noun and returns it if it's the correct type, and crashes if not. The mold here checks whether the input noun is a loobean (0/true or 1/false).

# 1079 3.7.1 Example: A Loobean Mold

```
1080
      :: loobean mold gate
1081
      > =loobean-mold ?
1082
1083
      :: below output truncated for our purposes here
1084
      > loobean-mold
1085
      < 1.fxf ... >
1086
1087
            grab the code for the battery
1088
      > -.loobean-mold
1089
      [8 \ [6 \ [5 \ [1 \ 0] \ \underline{0} \ 6] \ [1 \ 0] \ \underline{6} \ [5 \ [1 \ 1] \ \underline{0} \ 6] \ [1 \ 1] \ \underline{0} \ 0]
1090
        <u>8 [5 [0 14] 0 2] 0 6]</u>
1883
```

#### 1093 So we:

- 1. Assign the mold gate denoted by ? to the face loobean-mold.
- 2. Examine what's inside, seeing that the head is an arm.
- 3. Print out that source code by calling the head.

The below code acts as a formula which evaluates its sample, and returns it if it's a loobean or crashes otherwise. Let's see how that works.

```
1100 \quad \begin{bmatrix} \underline{6} & [\underline{5} & [\underline{1} & 0] & \underline{0} & 6 \end{bmatrix} \begin{bmatrix} \underline{1} & 0 \end{bmatrix} \underbrace{\underline{6}} \begin{bmatrix} \underline{5} & [\underline{1} & 1] & \underline{0} & 6 \end{bmatrix} \begin{bmatrix} \underline{1} & 1 \end{bmatrix} \underbrace{\underline{0}} \underbrace{0} \end{bmatrix}
```

We start with <u>6</u>, which means this is an if-else. The true/false test is the next element:

```
1103 [5 [1 0] 0 6]
```

1101

1102

1104

1105

1106

1107

1108

1109

1110

1111

1112

1113

1114

1115

1116

1117

1118

1120

1121

1122

1123

1124

1126

1127

1129

This compares the quoted value 0 (from  $[\underline{1} \ 0]$ ) with the value at memory slot 6 in the subject ( $[\underline{0} \ 6]$ ).

What is the subject and what is at mem slot 6? This code is the arm of a gate, so the subject is that gate/core: [battery [sample payload]]. We are looking at the battery right now, and so [0 6] yanks the head of the tail, or the sample!

We know the sample will be a noun that we're testing for loobeanness, so this code starts by seeing whether the sample is the value 0/true. If it is, the next element is  $[\underline{1} \ 0]$ , so we return 0 if the sample is 0.

Otherwise, we run the second branch of the if-else conditional:

```
[6 [5 [1 1] 0 6] [1 1] 0 0]
```

This is also an opcode  $\underline{6}$  if-else. Once again, it compares something to the value at mem slot 6 (the sample), but this time it checks whether that is the value 1. If it is, it runs the formula  $[\underline{1} \ 1]$  to return 1.

If not, it means our input was neither 0 nor 1 and is not a loobean, so we run the formula  $[\underline{0} \ 0]$ , which always crashes (as there is no memory slot 0).<sup>6</sup> This is exactly what we want—crash if the sample is not a loobean!

The second part of the formula which is composed with an  $\underline{8}$  is:

```
[<u>8</u> [<u>5</u> [<u>0</u> 14] <u>0</u> 2] <u>0</u> 6]
```

This checks whether the battery of the current gate is equivalent to the head of the context; if so, it returns the sample. This

<sup>&</sup>lt;sup>6</sup>Recall that in the binary tree addressing scheme of Nock, 1 refers to the entire noun, 2 to its head if it exists, and 3 to its tail if it exists.

is a artifact of a type promotion in Hoon which converts a bare 1130 loobean value to a constant by pinning a constant as an exam-1131 ple first (i.e., at slot 14). (It's completely extraneous in Nock 1132 itself, however.) 1133

#### 3.8 Summary

1134

1137

1141

1150

In this section, we've seen how to use almost all of the remain-1135 ing opcodes, which build Nock up to a slightly more expres-1136 sive level. We also saw how Hoon cores start to arise pretty naturally out of the Nock primitives, especially 8 and 9. Then we walked through real production Nock code to show that 1139 everything we've learned so far works exactly as expected in 1140 the wild. In the next section, you'll learn how to write real programs in Nock, and compose those programs to make new 1142 ones. Finally, hopefully you now see that, whatever its other 1143 limitations, Nock is not particularly obscurantist, and is fairly 1144 straightforward to parse, once you understand its syntax and idioms. 1146

#### Interlude 4

In this section, we cover some loose ends deriving from the 1148 previous discussion. 1149

#### 4.1 Order of Operations

At this point, we can answer the question of what order Nock 1151 evaluates expressions in. The answer is straightforward: it 1152 starts with code that has only one pseudocode operator and 1153 all Nock code, as below: 1154

```
*[50 <u>4 4</u> [5 [0 1] [<u>1</u> 50]]]
1155
```

The interpreter can then be thought of as moving left-to-right, expanding according to its rules, until it can't expand any further into pseudocode: 1158

```
*[50 <u>4</u> <u>4</u> [<u>5</u> [<u>0</u> 1] [<u>1</u> 50]]]
         +*[50 <u>4</u> [<u>5</u> [<u>0</u> 1] [<u>1</u> 50]]]
1160
```

```
++*[50 5 [0 1] [1 50]]
1161
     ++=[*[50 \ [0 \ 1]] \ *[50 \ [1 \ 50]]]
1162
     ++=[/[1 50] 50]
1163
     Once we get to that last line, ++=[/[1 50] *[50 [1 50]]],
1164
     no further expansion into pseudocode is possible.
1165
         At this point, the interpreter expands "inside-out", starting
1166
     from the deepest operators. In this case, those are /[1 50]
1167
     and 50, so it does those:
1168
     ++=[50 50]
1169
     Then it moves to the next-furthest-inside operator: =, and then
1170
     to the + operators:
1171
     ++=[50 50]
1172
     ++(0)
1173
     +(1)
1174
1175
```

# 4.1.1 Summary of Nock Order of Operations

- On the first pass, when we have Nock code \*(nock-code), we expand left-to-right into pseudocode.
  - 2. On the second pass, when we've fully expanded pseudocode, we evaluate operators from the inside-out.

# 4.2 11, Hints and Side Effects for the Interpreter

Opcode <u>11</u> lets you compute side effects and pass information to Nock's interpreter, for that interpreter to do what it wants. The most common uses are things like jets and debugging prints. In theory, this opcode is fairly dangerous, since it tells the interpreter that it's free to do anything.

There are two different versions of  $\underline{11}$ , depending on whether the head following 11 is an atom or a cell

```
1189 :: head is an atom (b)

1190 *[a <u>11</u> b c] *[a c]

1191

1192 :: head is a cell ([b c])

1193 *[a <u>11</u> [b c] d] *[[*[a c] *[a d]] <u>0</u> 3]
```

1176

1177

1178

1179

1180

1181

1182

1183

1184

1185

1186

1187

1188

This pseudocode lets <u>11</u> handle two separate use cases:

- 1. We just want the interpreter to process a message in its own language/environment.
- 2. We want the interpreter to compute some Nock code with Nock semantics before processing the hint.

#### 4.2.1 Option #1: Interpreter Processes a Message

1200 Imagine the following Nock:

```
1201 . * ([50 51] [<u>11</u> 369 <u>0</u> 2])
```

1195

1196

1197

1198

1199

1202

1203

1204

1205

1206

1207

1208

1211

1212

1214

1215

1216

This passes the value 369 to the interpreter, where maybe it would be the key in a hashmap. The value associated with the could be the function to debug print the whole operation. Or the value could be a jet function that specifically knows how to compute the  $\underline{0}$  opcode faster when it's followed by a  $\underline{2}$ .

After the interpreter does whatever, it needs to then discard the value 292.989, and computes

```
1209 > .*([50 51] [<u>0</u> 2])
1210 50
```

# 4.2.2 Option #2: Run Nock Code and then Process a Message

In this case, instead of passing a static value to the interpreter, we pass it a Nock computation against the subject. The result of that computation will then be available to the interpreter to use as a message.

#### 4.2.3 Practical Use of <u>11</u>: Jet Registration

1217 Jet registration in Hoon usually looks like this:

```
1218
1219 ++ my-function
1220 ~/ %my-function
1221 |= var=*
1222 ...
```

~/ "fassig" uses Nock <u>11</u> to pass the value %my-function to the interpreter, where it is associated with the C code that produces the product of my-function.

#### 4.3 Nock in Hoon

12/1

In Hoon, you can acquire the Nock code for any function by using the != "zaptis" rune. Since this produces Nock code complete with gate calls, you may also find it helpful to cast a data type as a noun using ^- \*, which will show you the raw Nock code for that value.

As you have seen, you can also evaluate raw Nock using the .\* "dottar" rune, which evaluates to Nock opcode  $\underline{2}$ . Other Nock runes in Hoon include .+, which is opcode  $\underline{4}$ ; .=, which is opcode  $\underline{5}$ ; and .?, which is opcode  $\underline{3}$ . Fake opcode  $\underline{12}$  is not actually Nock, but is used in a virtualization context to resolve values that may not be available in the current subject.

# 5 The Core as Design Pattern

At this point, you now know everything there is to know about Nock syntax and basic code manipulation. However, if I asked you to go write a simple program in Nock, you'd probably have a hard time getting started. It's really helpful to see some examples of how Nock programs are designed and patterns that tend to recur in them.

We can set variables to formulas in the Urbit Dojo. We will use that in this lesson to make code easy to follow, and to show how Nock programs are made out of discrete chunks.

```
1249 > =increment-formula [4 0 1]

1250 > .*(50 increment-formula)

1251 51

1252 > =increment-formula
```

# 5.1 Building a Core Manually

Every Nock program is just a formula that takes in a subject (argument). In order to run full programs against a subject, we follow the following pattern:

1. First pass: set up the dominos inside an opcode like  $\underline{2}$  (or  $\underline{7/8/9}$ ). This will create a core, and evaluate it on the second pass.

2. Second pass: evaluate that core to get our result.

The pseudocode for opcode  $\underline{2}$ , this looks like:

```
*[subject 2 core-formula call-formula]
```

Almost always, however, we'll use the  $\underline{9}$  opcode (which is sugar on top of  $\underline{2}$ ) to actually start our programs running, since it supports the concept of "cores" and "arms" very naturally. (For those who know Hoon, this "set up a core and run it" pattern should feel similar to  $| ^\circ$  or = >.)

### 5.1.1 Example: A Simple Increment Gate

Let's take code to increment the subject and turn it into a Hoonstyle gate (a core with one arm and a sample). The code itself is simple:  $[\underline{4} \ \underline{0} \ 1]$ .

Our code expects the subject to be an atom, but Hoon gates expect their subject to be the gate itself, which has the form [battery payload]. So we need to adjust our formula to accept this subject format.

We put our formula in the head, and put a default ("bunted") sample of 0 as the payload.

```
1287 > =inc-gate [inc 0]
1288 > inc-gate
1289 [[7 [0 3] 4 0 1] 0]
```

Arms are best invoked using the  $\underline{9}$  opcode. Here we use the "standard" version of it:  $\underline{[9 \ 2 \ \underline{0} \ 1]}$ . Recall that  $\underline{9}$  is  $\star \underline{[a \ \underline{9} \ b]}$  c], where c is the "new subject" formula, and b is the memory slot of that new subject to take an arm from. Here we use  $\underline{[0]}$  1] to keep our subject (the gate) unchanged.

```
1295 > .*(inc-gate [<u>9</u> 2 <u>0</u> 1])
1296 1
```

1298

1299

1314

1315

1316

1317

1318

1319

1320

1321

1322

1323

1324

1325

1326

1327

1328

1329

To "pass a parameter" to our gate, we use the <u>10</u> opcode to edit the subject (the inc-gate core) and replace the payload with a value. We make our subject [inc-gate val-to-increment]:

```
:: notice how opcode 10 can replace the last element
1300
           of inc-gate
      :: the tail of inc-gate was 0; now it's 36
1302
      > .*([inc-gate 36] [<u>10</u> [<u>3</u> [<u>0</u> 3]] <u>0</u> 2])
1303
      [[7 [0 3] 4 0 1] 36]
1305
           now use that to set up the subject with a new
1306
           tail (sample)
1307
           and call arm in memory slot 2 (our increment
1308
           formula)
      > .*([inc-gate 36] [9 2 10 [3 [0 3]] 0 2])
1310
1311
      > .*([inc-gate 562] [\underline{9} 2 \underline{10} [\underline{3} [\underline{0} 3]] \underline{0} 2])
1312
      563
1313
```

## 5.1.2 Summary

What was the point of this? We took a perfectly good increment formula that worked on atoms, and made it more complicated for the same result. Lame!

In fact, however, this setup will come in really handy when we have multiple arms in our core, not just one. We've made it easy to set up a clean environment for increment to find its sample in, no matter what other data is present. This will come in *very* handy in our third example program.

First, however, let's look at putting a more complicated formula inside a gate. (Hint: the process is the same).

# 5.1.3 Example: A Decrement Gate

Here we take a piece of code that decrements the subject when the subject is an atom, and turn it into a gate. For now, we will take it as a given that the decrement function given below works. (The code will crash when the input is a cell or 0)

```
> =dec [6 [5 [0 1] [1 0]] [0 0] [6 [3 0 1] [0 0] [8
1332
            [1 0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0 2] [4 0
1333
            6] <u>0</u> 7] <u>9</u> 2 <u>0</u> 1]]]
1334
1335
      :: testing in the Dojo
1336
      > .*(100 dec)
1333
1338
      99
1339
1340
      :: with non-atom input we crash
      .*([100 101] dec)
1341
      dojo: hoon expression failed
1349
           Let's re-jigger the decrement formula to accept the subject
1344
      in format [battery payload]:
1345
            gets value in mem slot 3 and applies the
1346
            decrement formula to it
1347
      > =dec-arm [7 [0 3] dec]
1348
1349
      :: verify that dec-arm decrements the value in the
1350
            tail
1351
      \rightarrow .*([1 88] dec-arm)
1352
      87
1353
1354
      :: output the full dec-arm formula code
1355
      > dec-arm
1356
      [7 \ [0 \ 3] \ \underline{6} \ [\underline{5} \ [0 \ 1] \ \underline{1} \ 0] \ [\underline{0} \ 0] \ \underline{6} \ [\underline{3} \ \underline{0} \ 1] \ [\underline{0} \ 0] \ \underline{8} \ [\underline{1}
1357
            0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0 2] [4 0 6]
1358
            0 7] 9 2 0 1]
1359
          Next, assemble the core:
1360
      :: all we need to do is add the sample! (we give it a
1361
            default value of 0)
1362
      > =dec-gate [dec-arm 0]
1363
      > dec-gate
      [[7 [0 3] 6 [5 [0 1] 1 0] [0 0] 6 [3 0 1] [0 0] 8 [1
1365
            0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0 2] [4 0 6]
1366
            0 7] 9 2 0 1] 0]
1367
       [ [ \underline{7} \ [\underline{0} \ 3] \ \underline{6} \ [\underline{5} \ [\underline{0} \ 1] \ \underline{1} \ 0] \ [\underline{0} \ 0] \ \underline{6} \ [\underline{3} \ \underline{0} \ 1] \ [\underline{0} \ 0] \ \underline{8} \ [\underline{1} 
1368
            0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0 2] [4 0 6]
1369
            0 7] 9 2 0 1] 0]
1370
```

As in the first example, we use the <u>10</u> opcode to edit the subject (the dec-gate core) and replace the payload with a value. We make our subject [dec-gate val-to-decrement].

```
confirm that our 10 code places 36 in the tail
1374
       > .*([dec-gate 36] [<u>10</u> [<u>3</u> [<u>0</u> 3]] <u>0</u> 2])
       [[7 [0 3] 6 [5 [0 1] 1 0] [0 0] 6 [3 0 1] [0 0] 8 [1
1376
             0] <u>8</u> [<u>1</u> 6 [<u>5</u> [<u>0</u> 7] <u>4</u> <u>0</u> 6] [<u>0</u> 6] <u>9</u> 2 [<u>0</u> 2] [<u>4</u> <u>0</u> 6]
1377
1378
             0 7] 9 2 0 1] 36]
1379
             now use that to set up the subject with a new
1389
             tail (sample)
1381
             and call arm in memory slot 2 (our decrement
1382
             formula)
       > .*([dec-gate 36] [\underline{9} 2 \underline{10} [\underline{3} [\underline{0} 3]] \underline{0} 2])
1384
1385
       > .*([dec-gate 562] [<u>9</u> 2 <u>10</u> [<u>3</u> [<u>0</u> 3]] <u>0</u> 2])
1386
       561
1387
```

## 5.1.4 Example: Comparing Two Numbers

In this example we will build a Nock core that compares two numbers and returns whether they are greater than, less than, or equal to each other.

The program specification is to take two numbers a and b and return:

```
• 0 if a = = b.
1395
• 1 if a > b.
1396
• 2 if a < b.</li>
```

1388

1392

1393

1397

1398

1399

1400

The trick is that the only mathematical operators we have for this are the 4 opcode for increment, 5 for equals, and a decrement Nock function that we will supply. The basic algorithm is:

```
    If a = = b, return 0.
    If a = = 0, return 2. (I.e, a < b).</li>
    If b = = 0, return 1. (I.e., a > b).
```

#### Recur with a and b both decremented.

The above works because if a is smaller than b, it hits o first. If b is smaller than a, it hits 0 first. If they are equal, we never decrement in the first place, so there are no issues with numbers going negative.

We will construct a core that has two arms: one arm with the algorithm logic, and a second arm with the decrement gate from the second example. These arms will live in the head of the core (think "battery").

We will put our variables a and b in the tail of the core (think "payload"). They will be updated before the core is called each time. (This is similar to a trap or door in Hoon).

The final core we will create will have the structure [battery payload], which can be broken down as:

```
[[main-logic dec-gate] a b]
```

1/10

Once the core is set up, we'll invoke the algorithm logic arm to set it running. We will build the main logic "inside-out", by first defining the "recur" code, and then inserting it into our if/else tests that implement the algorithm.

This recursion logic assumes the decrement gate lives in the tail of the battery (the battery is the head), i.e.  $[\underline{0} \ 5]$ . We are using  $[\underline{10} \ [\underline{3} \ [\underline{0} \ memory-slot]] \ \underline{0} \ 5]$  to extract the decrement gate from our core, and then we invoke it with  $[\underline{9} \ 2]$ 

This is identical to what we did in the second example—the only difference is that we now do it twice: once for memory slot 6 (a) and again for memory slot 7 (b).

```
> =dec-a [9 2 10 [3 [0 6]] 0 5]
1431
     \Rightarrow =dec-b [9 2 10 [3 [0 7]] 0 5]
1432
1433
         we can test our formulas by making a
1434
         dummy core (a = 33, b = 77)
1435
         first formula in battery just returns
1436
         the current subject
1437
     > =dummy-core [[[0 1] dec-gate] 33 77]
1438
     > .*(dummy-core dec-a)
1439
     32
1440
```

```
> .*(dummy-core dec-b)
1441
      76
1442
          The recursion formula here operates by invoking its subject—
1443
      itself—and applying the dec-a and dec-b formulas to it. Then
1444
      we use opcode 9 for the new subject's setup: replace payload
1445
      with new a and b run the current core
1446
      > =recur [9 4 [[0 2] dec-a dec-b]]
1447
1448
      :: test it -- we want to see same battery,
1449
      :: with payload being replaced with 32 and 76
1450
      > .*(dummy-core recur)
1453
      [[[\underline{0} \ 1] \ [\underline{7} \ [\underline{0} \ 3] \ \underline{6} \ [\underline{5} \ [\underline{0} \ 1] \ \underline{1} \ 0] \ [\underline{0} \ 0] \ \underline{6} \ [\underline{3} \ \underline{0} \ 1] \ [\underline{0}
1452
            0] 8 [1 0] 8 [1 6 [5 [0 7] 4 0 6] [0 6] 9 2 [0 2]
1453
            [<u>4</u> <u>0</u> 6] <u>0</u> 7] <u>9</u> 2 <u>0</u> 1] 0] 32 76]
1454
1455
           clean up the dummy-core
1456
      > = dummy-core
1457
          The main logic is then:
1458
1459
           check whether a is 0; return 2 if true
1460
         else check whether b is 0; return 1 if true
1461
          else recur
1462
      \rightarrow =main-logic [6 [5 [0 6] [1 0]] [1 2] [6 [5 [0 7] [1
1463
            0]] [1 1] recur]]
1464
1465
      :: test (we aren't doing the equality case here)
1466
            a < b
1467
      > .*([[main-logic dec-gate] 9 10] [9 4 0 1])
1468
1469
      :: a > b
1470
      > .*([[main-logic dec-gate] 10 9] [9 4 0 1])
1471
1473
          We build an outer test for whether a = = b, and then if it's
1474
      not, we use the 1 opcode to return the core.
1475
      > =battery [main-logic dec-gate]
1476
1477
            payload for our core is memory slots 2 and 3 (a
1478
            and b)
1479
      \rightarrow =comparison [\underline{6} [\underline{5} [\underline{0} 2] [\underline{0} 3]] [\underline{1} 0] [\underline{9} 4 [[\underline{1}
1480
            battery] [<u>0</u> 2] <u>0</u> 3]]]
1481
```

```
1488
1483 :: test our 3 cases
1484 > .*([0 8] comparison)
1485 2
1486 > .*([0 0] comparison)
1487 0
1488 > .*([8 0] comparison)
```

We can take this code and use it *anywhere* that our subject is a cell of two numbers, and it will "just work".

One common convention used in languages like Hoon is to denote Nock rules as constants, e.g. %0 (corresponding to our  $\underline{0}$ ). This makes it much more straightforward to interpret hand-rolled Nock code with addresses and constants present.

# 6 Conclusion

After this tutorial, you should be prepared to interpret Nock code when it is produced by Hoon. Although there is only rarely any call to write Nock code directly, you should now have a good understanding of how to read and interpret it. As you write your own interpreters in the future, you may even have call to hand-roll Nock expressions on occasion.